Problem 1 :

```
Clear["Global`*"]
```

```
ContourPlot[x^2 + y^2, {x, -1, 1}, {y, -1, 1}]
```




```
v = {x, y};
g = DiagonalMatrix[{1, 1}];
g // MatrixForm
```

$$\begin{pmatrix} 1 & 0 \\ 0 & 1 \end{pmatrix}$$

```
rot = {{Cos[θ], Sin[θ]}, {-Sin[θ], Cos[θ]}};
rot // MatrixForm
```

$$\begin{pmatrix} \text{Cos}[\theta] & \text{Sin}[\theta] \\ -\text{Sin}[\theta] & \text{Cos}[\theta] \end{pmatrix}$$

```
v2 = rot.v;
v2 // MatrixForm
```

$$\begin{pmatrix} x\,\text{Cos}[\theta] + y\,\text{Sin}[\theta] \\ y\,\text{Cos}[\theta] - x\,\text{Sin}[\theta] \end{pmatrix}$$

```
v.g.v
```

$x^2 + y^2$

```
v2.g.v2
```

$(y\,\text{Cos}[\theta] - x\,\text{Sin}[\theta])^2 + (x\,\text{Cos}[\theta] + y\,\text{Sin}[\theta])^2$

```
v2.g.v2 // Simplify
```

$x^2 + y^2$

Problem 2:

```
Clear["Global`*"]
```

In[◦]:=
```
v = {x, y};
g = {{1, 1}, {1, 1}};
g // MatrixForm
```

Out[◦]//MatrixForm=

$$\begin{pmatrix} 1 & 1 \\ 1 & 1 \end{pmatrix}$$

In[◦]:=
```
v.g.v // Simplify
```

Out[◦]= $(x + y)^2$

In[◦]:=
```
ContourPlot[(Abs[x] + Abs[y])^2, {x, 0, 1}, {y, 0, 1}]
```

Out[◦]=



```
ContourPlot[(Abs[x] + Abs[y])^2, {x, -1, 1}, {y, -1, 1}]
```



Problem 3:

```
Clear["Global`*"]
```

```
ContourPlot[t^2 - x^2, {t, -1, 1}, {x, -1, 1}]
```



```
v = {t, x};
g = DiagonalMatrix [{1, -1}];
g // MatrixForm
```

$$\begin{pmatrix} 1 & 0 \\ 0 & -1 \end{pmatrix}$$

```
rot = {{Cosh[θ], Sinh[θ]}, {+Sinh[θ], Cosh[θ]}};
rot // MatrixForm
```

$$\begin{pmatrix} \text{Cosh}[\theta] & \text{Sinh}[\theta] \\ \text{Sinh}[\theta] & \text{Cosh}[\theta] \end{pmatrix}$$

```
v2 = rot.v;
v2 // MatrixForm
```

$$\begin{pmatrix} t\,\text{Cosh}[\theta] + x\,\text{Sinh}[\theta] \\ x\,\text{Cosh}[\theta] + t\,\text{Sinh}[\theta] \end{pmatrix}$$

```
v.g.v
```

$t^2 - x^2$

```
v2.g.v2
```

$(-x\,\text{Cosh}[\theta] - t\,\text{Sinh}[\theta])\,(x\,\text{Cosh}[\theta] + t\,\text{Sinh}[\theta]) + (t\,\text{Cosh}[\theta] + x\,\text{Sinh}[\theta])^2$

```
v2.g.v2 // Simplify
```

$t^2 - x^2$

Problem 4:

```
Clear["Global`*"]
```

```
g = DiagonalMatrix [{1, -1, -1, -1}];
```

```
p1 = {e1, 0, 0, +p};
p2 = {e2, 0, 0, -p};
p12 = p1 + p2;
```

```
eqs =
 {p1.g.p1 == m12,
  p2.g.p2 == m22,
  p12.g.p12 == s}
```

$\{e1^2 - p^2 == m12,\ e2^2 - p^2 == m22,\ (e1 + e2)^2 == s\}$

```
Solve[eqs, {m12, m22, s}] // Transpose // TableForm
```

$m12 \to e1^2 - p^2$

$m22 \to e2^2 - p^2$

$s \to (e1 + e2)^2$

```
sol = Solve[eqs, {e1, e2, p}] // Simplify
```

$$\left\{\left\{e1 \to \frac{-m12 + m22 - s}{2\sqrt{s}}, e2 \to \frac{m12 - m22 - s}{2\sqrt{s}}, p \to -\frac{\sqrt{m12^2 + (m22 - s)^2 - 2\, m12\, (m22 + s)}}{2\sqrt{s}}\right\},\right.$$

$$\left\{e1 \to \frac{-m12 + m22 - s}{2\sqrt{s}}, e2 \to \frac{m12 - m22 - s}{2\sqrt{s}}, p \to \frac{\sqrt{m12^2 + (m22 - s)^2 - 2\, m12\, (m22 + s)}}{2\sqrt{s}}\right\},$$

$$\left\{e1 \to \frac{m12 - m22 + s}{2\sqrt{s}}, e2 \to \frac{-m12 + m22 + s}{2\sqrt{s}}, p \to -\frac{\sqrt{m12^2 + (m22 - s)^2 - 2\, m12\, (m22 + s)}}{2\sqrt{s}}\right\},$$

$$\left.\left\{e1 \to \frac{m12 - m22 + s}{2\sqrt{s}}, e2 \to \frac{-m12 + m22 + s}{2\sqrt{s}}, p \to \frac{\sqrt{m12^2 + (m22 - s)^2 - 2\, m12\, (m22 + s)}}{2\sqrt{s}}\right\}\right\}$$

```
(* TAKE THE SOLUTION THAT IS POSITIVE FOR BIG S *)
sol[[4]]
```

$$\left\{e1 \to \frac{m12 - m22 + s}{2\sqrt{s}}, e2 \to \frac{-m12 + m22 + s}{2\sqrt{s}}, p \to \frac{\sqrt{m12^2 + (m22 - s)^2 - 2\, m12\, (m22 + s)}}{2\sqrt{s}}\right\}$$

```
del2[a_, b_, c_] := a^2 + b^2 + c^2 - 2 (a b + b c + c a)
```

```
test1 = del2[m12, m22, s] / (4 s)
```

$$\frac{m12^2 + m22^2 + s^2 - 2\,(m12\, m22 + m12\, s + m22\, s)}{4\, s}$$

```
test2 = p^2 /. sol[[4]]
```

$$\frac{m12^2 + (m22 - s)^2 - 2\, m12\, (m22 + s)}{4\, s}$$

```
test1 == test2 // Simplify
```

True

Problem 5:
Start in CMS, equal mass particles

```
Clear["Global`*"]
```

```
g = DiagonalMatrix[{1, -1, -1, -1}];
```

```
p1 = {e, 0, 0, +p};
p2 = {e, 0, 0, -p};
p12 = p1 + p2;
```

```
eqs =
 {p1.g.p1 == m2,
  p2.g.p2 == m2,
  p12.g.p12 == s}
```

$\{e^2 - p^2 == m2, e^2 - p^2 == m2, 4e^2 == s\}$

Invariant mass before : = s = 4e^2

Invariant mass after:

```
p3 = {m, 0, 0, 0};
p4 = {m, 0, 0, 0};
p5 = {mHiggs, 0, 0, 0};
p345 = p3 + p4 + p5
```

{2 m + mHiggs, 0, 0, 0}

```
eq = p345.g.p345 == 4 e^2
```

$(2\,m + mHiggs)^2 == 4\,e^2$

```
sol = Solve[eq, e] // Simplify
```

$\left\{\left\{e \to -m - \frac{mHiggs}{2}\right\}, \left\{e \to m + \frac{mHiggs}{2}\right\}\right\}$

```
(* Take positive solution *)
sol[[2]]
```

$\left\{e \to m + \frac{mHiggs}{2}\right\}$

```
(* Take positive solution *)
sol[[2]] /. {m → 1, mHiggs → 126}
```

{e → 64}

Now in lab frame, equal mass particles

```
p1 = {e, 0, 0, +p};
p2 = {m, 0, 0, 0};
p12 = p1 + p2;
```

```
s == p12.g.p12 /. {p^2 → e^2 - m2} // Simplify
```

$2\,e\,m + m^2 + m2 == s$

```
(* Invariant mass after remains the same: *)
p345.g.p345
```

$(2\,m + mHiggs)^2$

```
eq2 = p345.g.p345 == p12.g.p12 //. {p^2 → e^2 - m2, m2 → m^2} // Simplify
```

$(2\,m + mHiggs)^2 == 2\,m\,(e + m)$

```
sol2 = Solve[eq2, e] // Simplify
```

$\left\{\left\{e \to m + 2\,mHiggs + \frac{mHiggs^2}{2\,m}\right\}\right\}$

```
e /. sol2[[1]] /. {m → 1, mHiggs → 126}
```

8191

Problem 6:

```
Clear["Global`*"]

g = DiagonalMatrix[{1, -1, -1, -1}];

p1 = {e, 0, 0, +p};
p2 = {e, 0, 0, -p};
p12 = p1 + p2;

b = {{γ, 0, 0, γ β}, {0, 1, 0, 0}, {0, 0, 1, 0}, {γ β, 0, 0, γ}};
b // MatrixForm
```

$$\begin{pmatrix} \gamma & 0 & 0 & \beta\,\gamma \\ 0 & 1 & 0 & 0 \\ 0 & 0 & 1 & 0 \\ \beta\,\gamma & 0 & 0 & \gamma \end{pmatrix}$$

```
eq = b.p2 == {m, 0, 0, 0} // Thread
```

$\{e\,\gamma - p\,\beta\,\gamma == m,\ True,\ True,\ -p\,\gamma + e\,\beta\,\gamma == 0\}$

```
sol = Solve[eq, {β, γ}] /. {e^2 - p^2 → m^2}
```

$\left\{\left\{\beta \to \frac{p}{e},\ \gamma \to \frac{e}{m}\right\}\right\}$

Now Check

```
b.p2 //. sol // Simplify
```

$\left\{\left\{\frac{e^2 - p^2}{m}, 0, 0, 0\right\}\right\}$

```
(b.p2 //. sol // Simplify) /. {p^2 → e^2 - m^2}
```

{{m, 0, 0, 0}}

```
(b.p1 //. sol // Simplify)
```

$\left\{\left\{\frac{e^2 + p^2}{m}, 0, 0, \frac{2\, e\, p}{m}\right\}\right\}$

Problem 7:

In[2]:=
```
Clear["Global`*"]
```

In[3]:=
```
F = {{0, Ex, Ey, Ez}, {-Ex, 0, Bz, -By}, {-Ey, -Bz, 0, Bx}, {-Ez, By, -Bx, 0}};
F // MatrixForm
```

Out[4]//MatrixForm=

$$\begin{pmatrix} 0 & Ex & Ey & Ez \\ -Ex & 0 & Bz & -By \\ -Ey & -Bz & 0 & Bx \\ -Ez & By & -Bx & 0 \end{pmatrix}$$

In[5]:=
```
B = {{γ, γ β, 0, 0}, {γ β, γ, 0, 0}, {0, 0, 1, 0}, {0, 0, 0, 1}};
B // MatrixForm
```

Out[6]//MatrixForm=

$$\begin{pmatrix} \gamma & \beta\gamma & 0 & 0 \\ \beta\gamma & \gamma & 0 & 0 \\ 0 & 0 & 1 & 0 \\ 0 & 0 & 0 & 1 \end{pmatrix}$$

In[12]:=
```
result = B.F.Transpose[B] // Simplify;
result // MatrixForm
```

Out[13]//MatrixForm=

$$\begin{pmatrix} 0 & -Ex\left(-1+\beta^2\right)\gamma^2 & (Ey + Bz\,\beta)\,\gamma & (Ez - By\,\beta)\,\gamma \\ Ex\left(-1+\beta^2\right)\gamma^2 & 0 & (Bz + Ey\,\beta)\,\gamma & -By\,\gamma + Ez\,\beta\,\gamma \\ -(Ey + Bz\,\beta)\,\gamma & -(Bz + Ey\,\beta)\,\gamma & 0 & Bx \\ -Ez\,\gamma + By\,\beta\,\gamma & (By - Ez\,\beta)\,\gamma & -Bx & 0 \end{pmatrix}$$

In[14]:=
```
rule = {γ^2 → 1/(1 - β^2)}
```

Out[14]= $\left\{\gamma^2 \to \frac{1}{1-\beta^2}\right\}$

In[18]:=
```
result /. rule // Simplify // MatrixForm
```

Out[18]//MatrixForm=

$$\begin{pmatrix} 0 & Ex & (Ey + Bz\,\beta)\,\gamma & (Ez - By\,\beta)\,\gamma \\ -Ex & 0 & (Bz + Ey\,\beta)\,\gamma & -By\,\gamma + Ez\,\beta\,\gamma \\ -(Ey + Bz\,\beta)\,\gamma & -(Bz + Ey\,\beta)\,\gamma & 0 & Bx \\ -Ez\,\gamma + By\,\beta\,\gamma & (By - Ez\,\beta)\,\gamma & -Bx & 0 \end{pmatrix}$$

To match book convention :

In[21]:=
```
result /. rule /. {β → -β} // Simplify // MatrixForm
```

Out[21]//MatrixForm=

$$\begin{pmatrix} 0 & Ex & (Ey - Bz\,\beta)\,\gamma & (Ez + By\,\beta)\,\gamma \\ -Ex & 0 & (Bz - Ey\,\beta)\,\gamma & -(By + Ez\,\beta)\,\gamma \\ -Ey\,\gamma + Bz\,\beta\,\gamma & -Bz\,\gamma + Ey\,\beta\,\gamma & 0 & Bx \\ -(Ez + By\,\beta)\,\gamma & (By + Ez\,\beta)\,\gamma & -Bx & 0 \end{pmatrix}$$

In[20]:=
```
(-1 + β^2) γ^2 /. rule // Simplify
```

Out[20]= -1